
Syntax of Mini-Pascal (Spring 2016)

Mini-Pascal is a simplified (and slightly modified) subset of Pascal. Generally, the meaning of the features of Mini-Pascal programs are similar to their semantics in other common imperative languages, such as C.

1. Mini-Pascal allows nested functions and procedures, i.e. subroutines which are defined within another. A nested function is invisible outside of its immediately enclosing function (procedure / block), but can access preceding visible local objects (data, functions, etc.) of its enclosing functions (procedures / blocks). Within the same scope (procedure / function / block), identifiers must be unique but it is OK to redefine a name in an *inner* scope.
2. A **var** parameter is passed *by reference*, i.e. its address is passed, and inside the subroutine the parameter name acts as a synonym for the variable given as an argument. A called procedure or function can freely read and write a variable that the caller passed in the argument list.
3. Mini-Pascal includes a C-style **assert** statement (if an assertion fails the system prints out a diagnostic message and halts execution).
4. The Mini-Pascal operation *a.size* only applies to values of type **array of T** (where *T* is a type). There are only one-dimensional arrays. Array types are compatible only if they have the same element type. Arrays' indices begin with zero. The compatibility of array indices and array sizes is usually checked at run time.
5. By default, variables in Pascal are not initialized (with zero or otherwise); so they may initially contain rubbish values.
6. A Mini-Pascal program can print numbers and strings via the predefined special routines *read* and *writeln*. The stream-style IO makes conversion of values from their text representation to their internal numerical (binary) representation.
7. Pascal is a case non-sensitive language, which means you can write the names of variables, functions and procedures in either case.
8. The Mini-Pascal *multiline comments* are enclosed within curly brackets and asterisks as follows: "{* ... *}".
9. Note that the names *Boolean*, *false*, *integer*, *read*, *real*, *size*, *string*, *true*, *writeln* are treated in Mini-Pascal as "predefined identifiers", i.e., it is allowed to use them as regular identifiers in Mini-Pascal programs.

The arithmetic operator symbols '+', '-', '*', and '/' represent the following functions, where *T* is either "*integer*" or "*real*".

```
"+" : (T, T) -> T           // addition
"- " : (T, T) -> T           // subtraction
"*"  : (T, T) -> T           // multiplication
"/"  : (T, T) -> T           // division
```

The operator '%' represents integer modulo operation. The operator '+' *also* represents string concatenation:

```
"%" : (integer, integer) -> integer    // integer modulo
"+" : (string, string) -> string       // string concatenation
```

The operators "**and**", "**or**", and "**not**" represent Boolean operations:

```
"or"  : (Boolean, Boolean) -> Boolean    // logical or
"and" : (Boolean, Boolean) -> Boolean    // logical and
"not" : (Boolean) -> Boolean             // logical not
```

The relational operators "=", "<>", "<", "<=", ">=", ">" are overloaded to represent the comparisons between two values of the same type, with the obvious meanings. They can be applied to values of the types *int*, *real*, *string*, *Boolean*.

Context-free syntax notation for Mini-Pascal

The syntax definition is given in so-called *Extended Backus-Naur* form (EBNF). In the following Mini-Pascal grammar, the use of curly brackets "{ ... }" means 0, 1, or more repetitions of the enclosed items. Parentheses may be used to group together a sequence of related symbols. Brackets ("[" "]") may be used to enclose optional parts (i.e., zero or one occurrence). Reserved keywords are marked bold (as "**bold**"). Operators, separators, and other single or multiple character tokens are enclosed within quotes (as ":="). Note that the syntax given below also specifies the precedence of operators (via productions defined at different hierarchical levels).

Context-free grammar

```
<program> ::= "program" <id> ";" <block> "."
<declaration> ::= "var" <id> { , <id> } ":" <type> |
                  "procedure" <id> "(" parameters ")" ";" <block> |
                  "function" <id> "(" parameters ")" ":" <type> ";" <block>
<parameters> ::= [ "var" ] <id> ":" <type> { , [ "var" ] <id> ":" <type> } | <empty>
<type> ::= <simple type> | <array type>
<array type> ::= "array" "[" [ <integer expr> ] "]" "of" <simple type>
<simple type> ::= <type id>
<block> ::= "begin" <statement> { ";" <statement> } [ ";" ] "end"
<statement> ::= <simple statement> | <structured statement> | <declaration>
<empty> ::=
```

```
<simple statement> ::= <assignment statement> | <call> | <return statement> |
                    <read statement> | <write statement> | <assert statement>
<assignment statement> ::= <variable> " := " <expr>
<call> ::= <id> "(" <arguments> ")"
<arguments> ::= expr { , expr } | <empty>
<return statement> ::= "return" [ expr ]
<read statement> ::= "read" "(" <variable> { , <variable> } ")"
<write statement> ::= "writeln" "(" <arguments> ")"
<assert statement> ::= "assert" "(" <Boolean expr> ")"
```

```
<structured statement> ::= <block> | <if statement> | <while statement>
<if statement> ::= "if" <Boolean expr> "then" <statement> |
                  "if" <Boolean expr> "then" <statement> "else" <statement>
<while statement> ::= "while" <Boolean expr> "do" <statement>
```

$\langle expr \rangle ::= \langle simple\ expr \rangle |$
 $\quad \langle simple\ expr \rangle \langle relational\ operator \rangle \langle simple\ expr \rangle$
 $\langle simple\ expr \rangle ::= [\langle sign \rangle] \langle term \rangle \{ \langle adding\ operator \rangle \langle term \rangle \}$
 $\langle term \rangle ::= \langle factor \rangle \{ \langle multiplying\ operator \rangle \langle factor \rangle \}$
 $\langle factor \rangle ::= \langle call \rangle | \langle variable \rangle | \langle literal \rangle | "(" \langle expr \rangle ")" | \textbf{not} \langle factor \rangle | \langle factor \rangle "." \textit{size}$
 $\langle variable \rangle ::= \langle variable\ id \rangle ["[" \langle integer\ expr \rangle "]"]$

$\langle relational\ operator \rangle ::= "=" | "<" | "<" | "<=" | ">=" | ">"$
 $\langle sign \rangle ::= "+" | "-"$
 $\langle adding\ operator \rangle ::= "+" | "- " | \textbf{or}$
 $\langle multiplying\ operator \rangle ::= "*" | "/" | "\%" | \textbf{and}$

Lexical grammar

$\langle id \rangle ::= \langle letter \rangle \{ \langle letter \rangle | \langle digit \rangle | "_" \}$
 $\langle literal \rangle ::= \langle integer\ literal \rangle | \langle real\ literal \rangle | \langle string\ literal \rangle$
 $\langle integer\ literal \rangle ::= \langle digits \rangle$
 $\langle digits \rangle ::= \langle digit \rangle \{ \langle digit \rangle \}$
 $\langle real\ literal \rangle ::= \langle digits \rangle "." \langle digits \rangle [\textbf{e} [\langle sign \rangle] \langle digits \rangle]$
 $\langle string\ literal \rangle ::= "\"" \{ \langle a\ char\ or\ escape\ char \rangle \} "\""$
 $\langle letter \rangle ::= a | b | c | d | e | f | g | h | i | j | k | l | m | n | o |$
 $\quad p | q | r | s | t | u | v | w | x | y | z | A | B | C |$
 $\quad D | E | F | G | H | I | J | K | L | M | N | O | P |$
 $\quad Q | R | S | T | U | V | W | X | Y | Z$
 $\langle digit \rangle ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9$
 $\langle special\ symbol\ or\ keyword \rangle ::= "+" | "-" | "*" | "\%" | "=" | "<" | "<" | ">" | "<=" | ">=" |$
 $\quad "(" | ")" | "[" | "]" | ":" | "." | "," | ";" | ":" | \textbf{or} |$
 $\quad \textbf{and} | \textbf{not} | \textbf{if} | \textbf{then} | \textbf{else} | \textbf{of} | \textbf{while} | \textbf{do} |$
 $\quad \textbf{begin} | \textbf{end} | \textbf{var} | \textbf{array} | \textbf{procedure} |$
 $\quad \textbf{function} | \textbf{program} | \textbf{assert}$
 $\langle predefined\ id \rangle ::= \textit{Boolean} | \textit{false} | \textit{integer} | \textit{read} | \textit{real} | \textit{size} | \textit{string} | \textit{true} | \textit{writeln}$
